
Title

mds — Multidimensional scaling for two-way data

Syntax

mds varlist
[

if
] [

in
]
, id(varname)

[
options

]

options description

Model
∗id(varname) identify observations
unit

[
(varlist2)

]
scale variables to min = 0 and max = 1

std
[
(varlist3)

]
scale variables to mean = 0 and sd = 1

measure(measure) similarity or dissimilarity measure (see [MV] measure option);
default is L2 (Euclidean)

s2d(standard) convert similarity to dissimilarity: dij =
√

sii + sjj − 2sij , the default
s2d(oneminus) convert similarity to dissimilarity: dij = 1 − sij

dimension(#) configuration dimensions; default is dimension(2)

addconstant make distance matrix positive definite

Reporting

neigen(#) maximum number of eigenvalues to display; default is neigen(10)

config display table with configuration coordinates
noplot suppress configuration plot

∗ id(varname) is required.
bootstrap, by, jackknife, rolling, statsby, and xi may be used with mds; see [U] 11.1.10 Prefix commands.
The maximum number of observations allowed in mds is the maximum matrix size; see [R] matsize.
See [MV] mds postestimation for features available after estimation.

Description
mds performs classical metric multidimensional scaling (MDS) for dissimilarity between observations

with respect to the variables in varlist. A wide selection of similarity and dissimilarity measures is
available, see the measure() option.

While mds computes dissimilarities from the observations, mdslong and mdsmat are for use
when you already have similarity or dissimilarity information. mdslong and mdsmat offer the same
statistical features but require different data organizations. mdslong expects the proximity information
in a “long format” (pairwise or dyadic form), while mdsmat performs MDS on a symmetric dissimilarity
matrix; see [MV] mdslong and [MV] mdsmat.

Options

�

� �

Model �

id(varname) is required; it specifies a variable that identifies observations. A warning message is
displayed if varname has duplicate values.

unit
[
(varlist2)

]
specifies variables that are transformed to min = 0 and max = 1 before entering

in the computation of similarities or dissimilarities. unit by itself, without an argument, is a
shorthand for unit( all).

Obviously, variables in unit() should not be included in std().

std
[
(varlist3)

]
specifies variables that are transformed to mean = 0 and sd = 1 before entering in

the computation of similarities or dissimilarities. std by itself, without an argument, is a shorthand
for std( all).

Obviously, variables in std() should not be included in unit().

measure(measure) specifies the similarity or dissimilarity measure. The default is measure(L2),
Euclidean distance. See [MV] measure option for detailed descriptions of the supported measures.

If a similarity measure is selected, the computed similarities will first be transformed into dissim-
ilarities, before proceeding with the scaling; see the s2d() option below.

Classical metric MDS with Euclidean distance is equivalent to principal component analysis (see
[MV] pca); the MDS configuration coordinates are the principal components.

s2d(conversion) specifies how measures in similarity form are converted to dissimilarities. The
following conversions are available:

standard dij =
√

sii + sjj − 2sij , the default

oneminus dij = 1 − sij

Obviously, s2d() should only be specified with measures in similarity form.

dimension(#) specifies the dimension of the approximating configuration. # defaults to 2 and should
not exceed the number of positive eigenvalues of the centered distance matrix.

addconstant specifies that if the double-centered distance matrix is not positive semi-definite (psd),
a constant should be added to the squared distances to make it psd and, hence, Euclidean. mds
adds the minimal constant, which is minus twice the smallest eigenvalue of the double-centered
distance matrix (Lingoes 1971).

�

� �

Reporting �

neigen(#) specifies the number of eigenvalues to be included in the table. The default is neigen(10).
Specifying neigen(0) suppresses the table.

config displays the table with the coordinates of the approximating configuration. This table may also
be displayed using the postestimation command estat config; see [MV] mds postestimation.

noplot suppresses the graph of the approximating configuration. Note that the graph can still be
produced later via mdsconfig, which also allows the standard graphics options for fine-tuning the
plot; see [MV] mds postestimation.

Remarks
Remarks are presented under the headings

Introduction
Euclidean distances
Non-Euclidean dissimilarity measures

Introduction

Multidimensional scaling (MDS) is a dimension-reduction and visualization technique. dissimilarities
(for instance, Euclidean distances) between observations in a high-dimensional space are represented
in a lower-dimensional space (typically two dimensions) so that the Euclidean distance in the lower-
dimensional space approximates the dissimilarities in the higher-dimensional space. See Kruskal and
Wish (1978) for a brief nontechnical introduction to MDS. Young and Hamer (1994) and Borg and
Groenen (1996) are more advanced textbook-sized treatments.

If you already have the similarities or dissimilarities of the n objects, you should continue by
reading [MV] mdsmat.

In many applications of MDS, however, the similarity or dissimilarity of objects is not measured,
but defined by the researcher in terms of variables (“attributes”) x1, . . . , xk that are measured on
the objects. The pairwise dissimilarity of objects can be expressed using a variety of similarity or
dissimilarity measures in the attributes (e.g., Mardia, Kent, and Bibby 1979, section 13.4; Cox and
Cox 2001, section 1.3). A common measure is the Euclidean distance L2 between the attributes of
the objects i and j,

L2ij =
√

(xi1 − xj1)2 + (xi2 − xj2)2 + · · · + (xik − xjk)2

A popular alternative is the L1 distance, also known as the cityblock or Manhattan distance. In
comparison to L2, L1 gives less influence to larger differences in attributes,

L1ij = |xi1 − xj1| + |xi2 − xj2| + · · · + |xik − xjk|

In contrast, we may also define the extent of dissimilarity between two observations as the maximum
absolute difference in the attributes and give a larger influence to larger differences,

Lij = max(|xi1 − xj1|, |xi2 − xj2|, · · · , |xik − xjk|)

These three measures are special cases of the Minkowski distance L(q), for q = 2 (L2), q = 1 (L1),
and q = ∞ (Linfinity), respectively. Minkowski distances with other values of q may be used, as
well. Stata supports a wide variety of other similarity and dissimilarity measures, both for continuous
variables and for binary variables. See [MV] measure option for details.

Multidimensional scaling constructs approximations for dissimilarities, not for similarities. Thus if
a similarity measure is specified, mds first transforms the similarities into dissimilarities. Two methods
to do this are available. The default standard method,

dissimij =
√

simii − 2simij + simjj

has a useful property: If the similarity matrix is positive semi-definite, a property satisfied by most
similarity measures, the standard dissimilarities are Euclidean.

In many applications, the number of observations exceeds the number of variables on which the
observations are compared, but this is not a requirement for MDS. MDS creates an n×n dissimilarity
matrix D from the n observations on k variables. It then constructs an approximation of D by the
Euclidean distances in a matching configuration Y of n points in p-dimensional space,

dissimilarity(xi, xj) ≈ L2(yi, yj) for all i, j

Typically, of course, p << k, and most often p = 1, 2, or 3.

A wide variety of MDS methods have been proposed. mds performs classical scaling that has its
roots in Young and Householder (1938) and Torgerson (1952). Classical scaling requires complete
and symmetric dissimilarity interval-level data.

Euclidean distances

Example 1

The most popular dissimilarity measure is Euclidean distance. We illustrate with data from table 7.1
of Yang and Trewn (2004, 182). This dataset consists of 8 variables with nutrition data on 25 breakfast
cereals.

. use http://www.stata-press.com/data/r9/cerealnut
(Cereal Nutrition)

. describe

Contains data from http://www.stata-press.com/data/r9/cerealnut.dta
obs: 25 Cereal Nutrition

vars: 9 24 Feb 2005 17:19
size: 1,150 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

brand str25 %25s Cereal Brand
calories int %9.0g Calories (Cal/oz)
protein byte %9.0g Protein (g)
fat byte %9.0g Fat (g)
Na int %9.0g Na (mg)
fiber float %9.0g Fiber (g)
carbs float %9.0g Carbs (g)
sugar byte %9.0g Sugar (g)
K int %9.0g K (mg)

Sorted by:

. summarize calories-K, sep(4)

Variable Obs Mean Std. Dev. Min Max

calories 25 109.6 21.30728 50 160
protein 25 2.68 1.314027 1 6

fat 25 .92 .7593857 0 2
Na 25 195.8 71.32204 0 320

fiber 25 1.7 2.056494 0 9
carbs 25 15.3 4.028544 7 22
sugar 25 7.4 4.609772 0 14

K 25 90.6 77.5043 15 320

. replace brand = subinstr(brand," ","_",.)
(20 real changes made)

We replaced spaces in the cereal brand names with underscores to avoid confusing which words in
the brand names are associated with which points in the graphs we are about to produce. Removing
spaces is not required.

The default dissimilarity measure used by mds is the Euclidean distance L2 computed on the raw data
(unstandardized). The summary of the 8 nutrition variables shows that K, Na, and calories—having
much larger standard deviations—will largely determine the Euclidean distances.

. mds calories-K, id(brand)

Classical metric multidimensional scaling
dissimilarity: L2, computed on 8 variables

Number of obs = 25
Eigenvalues > 0 = 8 Mardia fit measure 1 = 0.9603
Retained dimensions = 2 Mardia fit measure 2 = 0.9970

abs(eigenvalue) (eigenvalue)^2
Dimension Eigenvalue Percent Cumul. Percent Cumul.

1 158437.92 56.95 56.95 67.78 67.78
2 108728.77 39.08 96.03 31.92 99.70

3 10562.645 3.80 99.83 0.30 100.00
4 382.67849 0.14 99.97 0.00 100.00
5 69.761715 0.03 99.99 0.00 100.00
6 12.520822 0.00 100.00 0.00 100.00
7 5.7559984 0.00 100.00 0.00 100.00
8 2.2243244 0.00 100.00 0.00 100.00

MDS configuration

The default MDS configuration graph can be improved upon using the mdsconfig postestimation
command. We will demonstrate this in a moment. But first, we explain the output of mds.

mds has extracted two dimensions, which is the default action. To assess goodness of fit, the
two statistics proposed by Mardia are reported (see Mardia, Kent, and Bibby 1979, section 14.4).
The statistics are defined in terms of the eigenvalues of the double-centered distance matrix. If the
dissimilarities are truly Euclidean, all eigenvalues are non-negative. Look at the eigenvalues. We may
interpret these as the extent to which the dimensions account for dissimilarity between the cereals.
Depending on whether you look at the eigenvalues or squared eigenvalues, it takes two or three
dimensions to account for over 99% of the dissimilarity.

We can produce a prettier configuration plot using the mdsconfig command; see [MV] mds
postestimation for details.

. generate place = 3

. replace place = 9 if inlist(brand,"Rice_Krispies","Nut_&_Honey_Crunch",
> "Special_K","Raisin_Nut_Bran","Lucky_Charms")
(5 real changes made)

. replace place = 12 if inlist(brand,"Mueslix_Crispy_Blend")
(1 real change made)

. mdsconfig, autoaspect mlabvpos(place)

Cheerios

Cocoa_Puffs

Honey_Nut_Cheerios

Kix

Lucky_Charms

Oatmeal_Raisin_Crisp

Raisin_Nut_Bran

Total_Corn_Flakes

Total_Raisin_Bran

Trix

Wheaties_Honey_Gold

All−Bran

Apple_Jacks

Corn_Flakes

Corn_Pops
Mueslix_Crispy_Blend

Nut_&_Honey_Crunch

Nutri_Grain_Almond_Raisin

Nutri_Grain_Wheat

Product_19

Raisin_Bran

Rice_Krispies

Special_K

Life

Puffed_Rice

−
15

0
−

10
0

−
50

0
50

10
0

15
0

D
im

en
si

on
 2

−200 −100 0 100 200 300
Dimension 1

Classical MDS

MDS configuration

The mlabvpos() graph option allowed fine control over the placement of the cereal brand names.
We created a variable called place giving clock positions where the cereal names were to appear in
relation to the plotted point. We set these to minimize overlap of the names. We also requested the
autoaspect option to obtain better use of the graphing region while preserving the scale of the x-
and y-axes.

MDS has placed the cereals so that all the brands fall within a triangle defined by Product 19,
All-Bran, and Puffed Rice. You can examine the graph to see how close your favorite cereal is to
the other cereals.

But, as we saw from the variable summary, three of the eight variables are controlling the distances.
If we want to provide for a more equal footing for the eight variables, we can request that mds
compute the Euclidean distances on standardized variables. Euclidean distance based on standardized
variables is also known as the Karl Pearson distance. We obtain standardized measures with the option
std.

(Continued on next page)

. mds calories-K, id(brand) std noplot

Classical metric multidimensional scaling
dissimilarity: L2, computed on 8 variables

Number of obs = 25
Eigenvalues > 0 = 8 Mardia fit measure 1 = 0.5987
Retained dimensions = 2 Mardia fit measure 2 = 0.7697

abs(eigenvalue) (eigenvalue)^2
Dimension Eigenvalue Percent Cumul. Percent Cumul.

1 65.645395 34.19 34.19 49.21 49.21
2 49.311416 25.68 59.87 27.77 76.97

3 38.826608 20.22 80.10 17.21 94.19
4 17.727805 9.23 89.33 3.59 97.78
5 11.230087 5.85 95.18 1.44 99.22
6 8.2386231 4.29 99.47 0.78 99.99
7 .77953426 0.41 99.87 0.01 100.00
8 .24053137 0.13 100.00 0.00 100.00

Notice that it now takes more MDS retained dimensions to account for over 99% of the underlying
distances. For this example, we have still retained only two dimensions. We specified the noplot
option because we wanted to exercise control over the configuration plot using the mdsconfig
command. We generated a variable named pos that will help minimize cereal brand name overlap.
The xnegate option negates the data in the first dimension to give a more natural left to right
orientation.

. generate pos = 3

. replace pos = 5 if inlist(brand,"Honey_Nut_Cheerios","Raisin_Nut_Bran",
> "Nutri_Grain_Almond_Raisin")
(3 real changes made)

. replace pos = 8 if inlist(brand,"Oatmeal_Raisin_Crisp")
(1 real change made)

. replace pos = 9 if inlist(brand,"Corn_Pops","Trix","Nut_&_Honey_Crunch",
> "Rice_Krispies","Wheaties_Honey_Gold")
(5 real changes made)

. replace pos = 12 if inlist(brand,"Life")
(1 real change made)

. mdsconfig, autoaspect xnegate mlabvpos(pos)

MDS configuration

This configuration plot, based on the standardized variables, better incorporates all of the nutrition
data. If you are familiar with these cereal brands, it is easy to spot groups of similar cereals appearing
near each other. The bottom left corner has several of the most sweetened cereals. The brands
containing the word “Bran” all appear to the right of center. Rice Krispies and Puffed Rice are the
farthest to the left with the more sweetened Puffed Rice down near the sugary cereals.

Classical multidimensional scaling based on standardized Euclidean distances is actually equivalent
to a principal component analysis of the correlation matrix of the variables. See Mardia, Kent, and
Bibby (1979, section 14.3) for details.

We now demonstrate this property by doing a principal component analysis extracting the leading
two principal components. See [MV] pca for details.

. pca calories-K, comp(2)

Principal components/correlation Number of obs = 25
Number of comp. = 2
Trace = 8

Rotation: (unrotated = principal) Rho = 0.5987

Component Eigenvalue Difference Proportion Cumulative

Comp1 2.73522 .680583 0.3419 0.3419
Comp2 2.05464 .436867 0.2568 0.5987
Comp3 1.61778 .879117 0.2022 0.8010
Comp4 .738659 .270738 0.0923 0.8933
Comp5 .46792 .124644 0.0585 0.9518
Comp6 .343276 .310795 0.0429 0.9947
Comp7 .0324806 .0224585 0.0041 0.9987
Comp8 .0100221 . 0.0013 1.0000



320 mds — Multidimensional scaling for two-way data

Principal components (eigenvectors)

Variable Comp1 Comp2 Unexplained

calories 0.1992 -0.0632 .8832
protein 0.3376 0.4203 .3253

fat 0.3811 -0.0667 .5936
Na 0.0962 0.5554 .3408

fiber 0.5146 0.0913 .2586
carbs -0.2574 0.4492 .4043
sugar 0.2081 -0.5426 .2765

K 0.5635 0.0430 .1278

Notice that the proportion and cumulative proportion of the eigenvalues in the PCA match the
percentages from MDS. We will ignore the interpretation of the principal components, but move
directly to the principal coordinates, also known as the scores of the PCA. We make a plot of the first
and second scores, using the scoreplot command; see [MV] scoreplot. We specify the mlabel()
option to label the cereals and the mlabvpos() option for fine control over placement of the brand
names.

. replace pos = 11 if inlist(brand,"All-Bran")
(1 real change made)

. scoreplot, mlabel(brand) mlabvpos(pos)

Compare this PCA score plot with the MDS configuration plot. Apart from some differences in how
the graphs were rendered, they are the same.

Non-Euclidean dissimilarity measures

With non-Euclidean dissimilarity measures, the parallel between PCA and MDS no longer holds.

Example 2

We illustrate MDS with non-Euclidean distance measures, we will analyze books on multivariate
statistics. Gifi (1990) reports on the number of pages devoted to six topics in 20 textbooks on
multivariate statistics. We added similar data on five more recent books.

. use http://www.stata-press.com/data/r9/mvstatsbooks

. describe

Contains data from http://www.stata-press.com/data/r9/mvstatsbooks.dta
obs: 25

vars: 8 15 Mar 2005 16:27
size: 825 (99.9% of memory free) (_dta has notes)

storage display value
variable name type format label variable label

author str17 %17s
math int %9.0g math other than statistics

(e.g., linear algebra)
corr int %9.0g correlation and regression,

including linear structural and
functional equations

fact byte %9.0g factor analysis and principal
component analysis

cano byte %9.0g canonical correlation analysis
disc int %9.0g discriminant analysis,

classification, and cluster
analysis

stat int %9.0g statistics, incl. dist. theory,
hypothesis testing & est.;
categorical data

mano int %9.0g manova and the general linear
model

Sorted by:

A brief description of the topics is given in the variable labels. For more details, we refer to Gifi
(1990, 15). Here are the data.

(Continued on next page)

. list, noobs

author math corr fact cano disc stat mano

Roy57 31 0 0 0 0 164 11
Kendall57 0 16 54 18 27 13 14
Kendall75 0 40 32 10 42 60 0
Anderson58 19 0 35 19 28 163 52

CooleyLohnes62 14 7 35 22 17 0 56

CooleyLohnes71 20 69 72 33 55 0 32
Morrison67 74 0 86 14 0 84 48
Morrison76 78 0 80 5 17 105 60

VandeGeer67 74 19 33 12 26 0 0
VandeGeer71 80 68 67 15 29 0 0

Dempster69 108 48 4 10 46 108 0
Tasuoka71 109 13 5 17 39 32 46
Harris75 16 35 69 24 0 26 41

Dagnelie75 26 86 60 6 48 48 28
GreenCaroll76 290 10 6 0 8 0 2

CailliezPages76 184 48 82 42 134 0 0
Giri77 29 0 0 0 41 211 32

Gnanadesikan77 0 19 56 0 39 75 0
Kshirsagar78 0 22 45 42 60 230 59
Thorndike78 30 128 90 28 48 0 0

MardiaKentBibby79 34 28 68 19 67 131 55
Seber84 16 0 59 13 116 129 101

Stevens96 23 87 67 21 30 43 249
EverittDunn01 0 54 65 0 56 20 30

Rencher02 38 0 71 19 105 135 131

For instance, the 1979 book by Mardia, Kent, and Bibby has 34 pages on mathematics (mostly linear
algebra), 38 pages on regression and related topics (in this particular case: simultaneous equations),
etc. In most of these books some pages are not classified. Anyway, the number of pages and the
amount of information per page vary widely between the books. A Euclidean distance measure is
not appropriate here. Standardization does not help us here—the problem is not differences in the
scales of the variables (columns), but those in the observations (rows). One possibility is to transform
the data into compositional data by dividing the variables by the total number of classified pages.
See Mardia, Kent, and Bibby (1979, 377–380) for a discussion of specialized dissimilarity measures
for compositional data. However, we can also use the correlation between observations (not between
variables) as the similarity measure. The higher the correlation between the attention given to the
various topics, the more similar two textbooks are. We do a classical MDS, suppressing the plot to
first assess the quality of a two-dimensional representation.

. mds math-mano, id(author) measure(corr) noplot

Classical metric multidimensional scaling
similarity: correlation, computed on 7 variables

dissimilarity: sqrt(2(1-similarity))

Number of obs = 25
Eigenvalues > 0 = 6 Mardia fit measure 1 = 0.6680
Retained dimensions = 2 Mardia fit measure 2 = 0.8496

abs(eigenvalue) (eigenvalue)^2
Dimension Eigenvalue Percent Cumul. Percent Cumul.

1 8.469821 38.92 38.92 56.15 56.15
2 6.0665813 27.88 66.80 28.81 84.96

3 3.8157101 17.53 84.33 11.40 96.35
4 1.6926956 7.78 92.11 2.24 98.60
5 1.2576053 5.78 97.89 1.24 99.83
6 .45929376 2.11 100.00 0.17 100.00

Again the quality of a two dimensional approximation is somewhat unsatisfactory with 67% and
85% of the variation accounted for according to the two Mardia criteria. Still, let’s look at the plot,
using a title that refers to the self-referential aspect of the analysis (Smullyan 1986). We reposition
some of the author labels to enhance readability using the mlabvpos() graph option.

. gen spot = 3

. replace spot = 2 if inlist(author,"Seber84","Kshirsagar78","Kendall75")
(3 real changes made)

. replace spot = 5 if author=="MardiaKentBibby79"
(1 real change made)

. replace spot = 9 if inlist(author, "Dagnelie75","Rencher02",
> "GreenCaroll76","EverittDunn01","CooleyLohnes62","Morrison67")
(6 real changes made)

. mdsconfig, mlabvpos(spot) title(This plot needs no title)

This plot needs no title

A striking characteristic of the plot is that the textbooks seem to be located on a circle. This is
a phenomenon that is regularly encountered in multidimensional scaling and was labeled the “horse
shoe effect” by Kendall (1971, 215–251). This phenomenon seems to occur especially in situations
in which a one-dimensional representation of objects needs to be constructed, e.g., in seriation



324 mds — Multidimensional scaling for two-way data

applications, from data in which small dissimilarities were measured accurately but moderate and
larger dissimilarities are “lumped together”.

❑ Technical Note
These data could also be analyzed in different ways. A particularly interesting method is corre-

spondence analysis (CA), which seeks a simultaneous geometric representation of the rows (textbooks)
and columns (topics). We used camat to analyze these data. The results for the textbooks were not
very different. Textbooks that were mapped as similar using MDS were also mapped this way by CA.
The Green and Carroll book that appeared much different from the rest was also displayed away from
the rest by CA; in the CA biplot, it was immediately clear that this book was so different because its
pages were predominantly classified by Gifi (1990) as mathematical. But CA also located the topics
in this space. The pattern was easy to interpret and not unexpected. The seven topics were mapped in
three groups. math and stat appear as two groups by themselves, and the five applied topics were
mapped close together. See [MV] ca for information on the ca command.

❑

Saved Results
mds saves in e():

Scalars
e(N) number of observations
e(p) number of dimensions in the approximating configuration
e(np) number of strictly positive eigenvalues
e(addcons) constant added to squared dissimilarities to force positive semi-definiteness
e(mardia1) Mardia measure 1
e(mardia2) Mardia measure 2

Macros
e(cmd) mds

e(method) classical

e(varlist) variables used in computing similarities or dissimilarities
e(id) observation identifier variable
e(numid) 1 if e(id) is numeric, 0 otherwise
e(dname) similarity or dissimilarity measure name
e(dtype) similarity or dissimilarity

e(s2d) standard or oneminus

e(unique) 1 if eigenvalues are distinct, 0 otherwise
e(properties) nob noV eigen

e(predict) mds p

e(estat cmd) mds estat

Matrices
e(D) dissimilarity matrix
e(Y) approximating configuration coordinates
e(Ev) eigenvalues
e(coding) variable standardization values; first column has value

to subtract and second column has divisor
e(linearf) two element vector defining the linear transformation; distance

equals first element plus second element times dissimilarity

Functions
e(sample) marks estimation sample

Methods and Formulas
mds is implemented as an ado-file.

mds creates a dissimilarity matrix D according to the measure specified in option measure(). See
[MV] measure option for descriptions of these measures. Subsequently, mds uses mdsmat to compute
the classical MDS solution for D. See [MV] mdsmat for information on the methods and formulas.

Also See

Complementary: [MV] mds postestimation

Related: [MV] biplot, [MV] ca, [MV] factor, [MV] mdslong, [MV] mdsmat, [MV] pca

Background: [U] 11.1.10 Prefix commands,

[U] 20 Estimation and postestimation commands

